See: https://github.com/fdformula/CalculusLabs/blob/main/text/Unit%2010%20-%20Parametric%20and%20Polar%20Curves.nb

- **Example 6 -** Animating cycloid

A cycloid is a plane curve traced out by a point P on the circumference of a circle of radius r as the circle rolls along a horizontal straight line.

Refer to the following figure. Let θ denote the angle ∠PQS. Them, $|\overline{\text{OS}}| = |\overset{\frown}{\text{PS}}| = r\,\theta$, and the parametric equations of point P are

$$x = |\overline{\text{OT}}| = |\overline{\text{OS}}| - |\overline{\text{TS}}| = r\,\theta - r\sin\theta = r\,(\theta - \sin\theta),$$
$$y = |\overline{\text{PT}}| = |\overline{\text{RS}}| = |\overline{\text{QS}}| - |\overline{\text{QR}}| = r - r\cos\theta = r\,(1 - \cos\theta).$$

In brief,

$$x = r\,(\theta - \sin\theta),\ \ y = r\,(1 - \cos\theta), \text{ where } -\infty < \theta < \infty.$$



In[1]:=
```
Manipulate[
 Module[{x, y, rPoint, P}, (* xMax: length of horizontal line *)
  P[r_, θ_] := {r (θ - Sin[θ]), r (1 - Cos[θ])};
  rPoint = 1.5 (xMax / 100.); (* radius of point P,
  relative to the scale of the horizontal line *)

  Show[{
    (* horizontal line and two walls at its ends *)
    Plot[0, {x, -r, xMax}, PlotRange → {-rPoint, 2 r + rPoint}, Ticks → None, Axes → None,
     AspectRatio → Automatic, PlotStyle → {Black, Thin}, ImageSize → 400],
    Graphics[{Black, Thin, Line[{{x, r}, {x, 0}}]}] /. {{x → -r}, {x → xMax}},

    ParametricPlot[P[r, t], {t, 0, θ}], (* cycloid *)

    (* rolling circle *)
    {x, y} = {r * θ, r}; (* center of the circle *)
    ParametricPlot[{x + r * Cos[t], y + r * Sin[ t]}, {t, 0, 2 Pi}, PlotStyle → Orange],
    Graphics[{Orange, Line[{P[r, θ], {x, y}}]}], (* more dynamic *)
    (* Table[
     Graphics[{LightOrange,Line[{{x+r*Cos[t],y+r*Sin[t]},{x-r*Cos[t],y-r*Sin[t]}}]}],
     {t,0,2π,π/12}], (* wheel *) *)

    Graphics[{Red, PointSize[Large], Point[P[r, θ]]}] (* point P *)
   }]
 ],
 {{θ, 0.0001}, 0.0001, (xMax - r) / r},
 {{r, 7.3 * (xMax / 100), "radius"}, 1, xMax / 2},
 {xMax, 100, 100, 100, ControlType → None}
 (* This special use of xMax is a trick,
 somehow ugly, to get rid of using a global variable,
 due to the constraint of Mathematica on Animate and Manipulate *)
]
```

Out[1]=



- **Example 7 -** Animating hypocycloid

A hypocycloid is a plane curve traced out by a fixed point *P* on a circle *C* of radius *b* as *C* rolls without slipping on the inside of a circle with center *O* and radius *a*, where $b < a$.

Refer to the graph. The dashed lines are either horizontal or vertical. Let $\theta = \angle ROS$ and $\alpha = \angle RQP$. Then, $|\overset{\frown}{PR}| = |\overset{\frown}{SR}|$, i.e., $b\,\alpha = a\,\theta$. Thus, $\alpha = \frac{a}{b}\theta$. In addition, $\angle QPT = \angle CQP = \angle RQP - \angle RQC = \alpha - \theta = \frac{a-b}{b}\theta$. Therefore, using θ as the parameter, the parametric equations of P are

$$x = |\overline{OB}| = \left|\overline{OA}\right| + |\overline{AB}| = |\overline{OQ}|\cos\theta + |\overline{TP}| =$$
$$(a-b)\cos\theta + |\overline{QP}|\ \cos\angle QPT = (a-b)\cos\theta + b\cos\left(\frac{a-b}{b}\theta\right),$$
$$y = |\overline{PB}| = |\overline{TA}| = \left|\overline{QA}\right| - |\overline{QT}| = |\overline{OQ}|\sin\theta - |\overline{QP}|\sin\angle QPT = (a-b)\sin\theta - b\sin\left(\frac{a-b}{b}\theta\right).$$



In summary,

$$x = (a-b)\cos\theta + b\cos\left(\frac{a-b}{b}\theta\right),\ y = (a-b)\sin\theta - b\sin\left(\frac{a-b}{b}\theta\right), \text{ where } -\infty < \theta < \infty.$$

```
Manipulate[
 Module[{P, x, y},
  P[θ_, a_, b_] := Module[{d, q}, d = a - b;
    q = d / b;
    {d * Cos[θ] + b * Cos[q * θ], d * Sin[θ] - b * Sin[q * θ]}];

  Show[{
    (* set up the canvas large enough to allow b > a *)
    ParametricPlot[{(a + b) Cos[t], (a + b) Sin[t]}, {t, 0, 2 Pi}, Ticks → None,
     AspectRatio → Automatic, PlotStyle → White, ImageSize → 170],

    (* the static circle *)
    ParametricPlot[{a Cos[t], a Sin[t]}, {t, 0, 2 Pi}, PlotStyle → {Thick, Orange}],

    ParametricPlot[P[t, a, b], {t, 0, θ}], (* hypocycloid *)

    (* the rolling circle *)
    {x, y} = {(a - b) Cos[θ], (a - b) Sin[θ]}; (* center of the cicle *)
    ParametricPlot[{x + b Cos[t], y + b Sin[t]}, {t, 0, 2 Pi}, PlotStyle → {Red, Thick}],
    Graphics[{Red, Line[{P[θ, a, b], {x, y}}]}], (* more dynamic *)

    Graphics[{Red, PointSize[Large], Point[P[ θ, a, b]]}] (* point P *)
   }]
 ],
 {{θ, π / 4}, 0.0001, 10 Pi, 0.1},
 {{a, 5}, 1, 10}, (* a=5, b=1; a=5, b=1.1; a=2, b=3; *)
 {{b, 1.1}, 1, 10}]
```



Play the animation for fun!

- **Example 8** - Animating epicycloid

A epicycloid is a plane curve traced out by a fixed point P on a circle C of radius b as C rolls without slipping on the outside of a circle with center O and radius a, where $b < a$.



Refer to the graph. Let $\theta = \angle QOS$ and $\alpha = \angle OQP$. Then, $|\overset{\frown}{PR}| = |\overset{\frown}{SR}|$, i.e., $b\,\alpha = a\,\theta$. Thus, $\alpha = \frac{a}{b}\,\theta$. Therefore, using θ as the parameter, the parametric equations of P are

$$x = |\overline{AT}| = |\overline{AQ}| - |\overline{TQ}| = |\overline{OQ}|\cos\theta - b\cos(\alpha+\theta) = (a+b)\cos\theta - b\cos\left(\frac{a+b}{b}\,\theta\right),$$
$$y = |\overline{PB}| = |\overline{TB}| - |\overline{TP}| = |\overline{AO}| - |\overline{TP}| = |\overline{OQ}|\sin\theta - b\sin(\alpha+\theta) = (a+b)\sin\theta - b\sin\left(\frac{a+b}{b}\,\theta\right).$$

In brief, the motion of any point P on the curve is governed by the parametric equations

$$x = (a+b)\cos\theta - b\cos\left(\frac{a+b}{b}\,\theta\right),\ y = (a+b)\sin\theta - b\sin\left(\frac{a+b}{b}\,\theta\right), \text{ where } 0 \le \theta < \infty.$$

```
Manipulate[
 Module[{P, x, y},
  P[θ_, a_, b_] := Module[{s, q}, s = a + b;
    q = s / b;
    {s * Cos[θ] - b * Cos[q * θ], s * Sin[θ] - b * Sin[q * θ]}];

  Show[{
    (* set up the canvas large enough to allow b > a *)
    ParametricPlot[{(a + 2 b) Cos[t], (a + 2 b) Sin[ t]}, {t, 0, 2 Pi}, Ticks → None,
     AspectRatio → Automatic, PlotStyle → White, ImageSize → 170],

    (* the static circle *)
    ParametricPlot[{a Cos[t], a Sin[ t]}, {t, 0, 2 Pi}, PlotStyle → {Thick, Orange}],

    ParametricPlot[P[t, a, b], {t, 0, θ}], (* epicycloid *)

    (* the rolling circle *)
    {x, y} = {(a + b) Cos[θ], (a + b) Sin[θ]}; (* the center *)
    ParametricPlot[{x + b Cos[t], y + b Sin[ t]}, {t, 0, 2 Pi}, PlotStyle → {Red, Thick}],
    Graphics[{Red, Line[{P[θ, a, b], {x, y}}]}], (* more dynamic *)

    (* point P *)
    Graphics[{Red, PointSize[Large], Point[P[θ, a, b]]}]}]
 ],
 {{θ, π / 4}, 0.0001, 20 Pi, 0.1},
 {{a, 5}, 1, 10}, (* a=5, b=1; a=5, b=1.1; a=2, b=3; *)
 {b, 1, 10}]
```



■ **Example 9**

Refer to the following figure. A stick of length L can smoothly slide through a point $B(b, 0)$. Provided one end of the stick can only move along a circle with center O and radius R, find parametric equations of the path of the other end of the stick.



Use ∠AOB as the parameter. Draw auxiliary lines $\overline{OA}$, $\overline{AD}$, and $\overline{PC}$, where the latter two are each perpendicular to the x-axis. Obviously,

$$|\overline{AD}| = R\sin\theta,$$
$$|\overline{OD}| = R\cos\theta,$$
$$|\overline{DB}| = |\overline{OB}| - |\overline{OD}| = b - R\cos\theta,$$
$$|\overline{BC}| = |\overline{OC}| - |\overline{OB}| = x - b,$$
$$|\overline{PC}| = -y.$$

By the Pythagorean Theorem,

$$L_1^2 = |\overline{AD}|^2 + |\overline{DB}|^2 = (R\sin\theta)^2 + (b - R\cos\theta)^2 = R^2 + b^2 - 2\,b\,R\cos\theta \cdots\cdots\cdots (1)$$

Since $L_1 + L_2 = L$, we have $L_2 = L - L_1$.

For △ABD and △PBC are similar triangles,

$$\frac{|\overline{DB}|}{L_1} = \frac{|\overline{BC}|}{L_2}, \text{ or } \frac{b - R\cos\theta}{L_1} = \frac{x-b}{L-L_1}$$
$$\frac{|\overline{AD}|}{L_1} = \frac{|\overline{PC}|}{L_2}, \text{ or } \frac{R\sin\theta}{L_1} = \frac{-y}{L-L_1}$$

Therefore,

$$x = b + \frac{L_2}{L_1}(b - R\cos\theta) = b + \left(\frac{L}{L_1} - 1\right)(b - R\cos\theta) \cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots (2)$$
$$y = -R\left(\frac{L_2}{L_1}\right)\sin\theta = R\left(1 - \frac{L}{L_1}\right)\sin\theta \cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots (3)$$

Now, combine (1), (2), and (3), we obtain

$$x = b + \left(\frac{L}{\sqrt{R^2+b^2-2\,b\,R\cos\theta}} - 1\right)(b - R\cos\theta),\ y = R\left(1 - \frac{L}{\sqrt{R^2+b^2-2\,b\,R\cos\theta}}\right)\sin\theta, \text{ where } -\infty < \theta < \infty.$$

How to play with the following animation? Click and drag the big blue point along the blue circle either clockwise or counterclockwise.

```
Manipulate[
 Module[
  {R = 1, b = 2, L = 3, tmp, ymax, redPoint, f, g, theta}, (* Make sure L ≥ R + b *)
  If[L < R + b, Abort[]];

  ymax = (L - (b - R)) Sin[π / 4]; (* estimated value *)
  theta = ArcCos[bluePoint[[1]] / Norm[bluePoint]];
  If[bluePoint[[2]] < 0, theta = 2 π - theta]; (* 3rd or 4th quadrant *)
  bluePoint = R {Cos[theta], Sin[theta]}; (* draw it to the circle always *)
  tmp = L / Sqrt[R^2 + b^2 - 2 b * R * Cos[theta]];
  redPoint = {b + (tmp - 1) (b - R * Cos[theta]), R (1 - tmp) Sin[theta]};

  Show[{
    Plot[{x}, {x, -R - 0.1, R + L + 0.1}, PlotRange → {-ymax, ymax},
     AspectRatio → Automatic, PlotStyle → White, Ticks → None, ImageSize → 350],
    (* setup proper canvas so that the frame will not change during animation *)

    ParametricPlot[{R Cos[θ], R Sin[θ]}, {θ, 0, 2 Pi}],
    Graphics[{Blue, PointSize[Large], Point[bluePoint]}],
    Graphics[{Black, PointSize[Large], Point[{b, 0}]}],
    Graphics[{Red, PointSize[Large], Point[redPoint]}],
    Graphics[{Black, Line[{redPoint, bluePoint}]}],
    Graphics[{Black, Text[O, {-0.15, -0.15}]}],
    Graphics[{Black, Text[B, {b, -0.2}]}],
    ParametricPlot[Module[{},
      tmp = L / Sqrt[R^2 + b^2 - 2 b * R * Cos[θ]];
      {b + (tmp - 1) (b - R * Cos[θ]), R (1 - tmp) Sin[θ]}], {θ, 0, 2 π}, PlotStyle → Red]
   }]], {{bluePoint, {1, 1}}, Locator}, Alignment → Center]
(* bluePoint doesn't have to be initialized
 to be on the circle since it will be drawn to it *)
```